
コマンドでジャーナルファイルにある特定のグローバル変数を検索す
る方法
これは InterSystems FAQ サイトの記事です。

%SYS.Journal.File クラスの ByTimeReverseOrder クエリ と %SYS.Journal.Record クラスの List
クエリを使用して検索することができます。

それぞれのクエリの役割は以下の通りです。

 

A) %SYS.Journal.File クラスの ByTimeReverseOrder クエリ

ジャーナルファイル名を取得できます。ジャーナルファイル名の降順で結果が返ります。

USER>set stmt=##class(%SQL.Statement).%New()

USER>set status=stmt.%PrepareClassQuery("%SYS.Journal.File","ByTimeReverseOrder")

USER>set rs=stmt.%Execute()

USER>while rs.%Next() { write rs.%Get("Name"),! }
c:\intersystems\irishcom\mgr\journal\20230725.002
c:\intersystems\irishcom\mgr\journal\20230725.001
C:\InterSystems\IRISHCom\mgr\journal\20230707.002

バージョン2018.1未満のバージョンでは以下の方法をご利用ください。

USER>set rs=##class(%ResultSet).%New("%SYS.Journal.File:ByTimeReverseOrder")

USER>do rs.Execute()

USER>while rs.Next() { write rs.Name,! }
c:\intersystems\cache\mgr\journal\20190620.003
c:\intersystems\cache\mgr\journal\20190620.002
c:\intersystems\cache\mgr\journal\20190620.001
c:\intersystems\cache\mgr\journal\20190610.001

 

 

B) %SYS.Journal.Record クラスの List クエリ

特定のジャーナルファイルのジャーナルレコードを取得できます。

バージョン2022.2以降でお試しいただく場合は、%SQL.Statementクラスをご利用ください。

USER>set status=stmt.%PrepareClassQuery("%SYS.Journal.Record","List")
 
USER>set rs2=stmt.%Execute("c:\intersystems\irishcom\mgr\journal\20230725.002")
 
USER>while rs2.%Next() {if rs2.%Get("GlobalReference")["TEST" write rs2.%Get(
"GlobalReference"),!}
^["^^c:\intersystems\irishcom\mgr\user\"]TEST(1)
     ??? ?? ???
^["^^c:\intersystems\irishcom\mgr\user\"]TEST(4)
^["^^c:\intersystems\irishcom\mgr\user\"]TEST(5)

バージョン2022.2未満でお試しいただく場合は、%ResultSetクラスをご利用ください。

USER>set rs2=##class(%ResultSet).%New("%SYS.Journal.Record:List")

USER>do rs2.Execute("c:\intersystems\irishcom\mgr\journal\20230725.002")

USER>while rs2.Next() { if rs2.Get("GlobalReference")["TEST" write rs2.Get(
"GlobalReference"),!}
^["^^c:\intersystems\cache\mgr\user\"]TEST(1)
     ??? ?? ???
^["^^c:\intersystems\cache\mgr\user\"]TEST(9)
^["^^c:\intersystems\cache\mgr\user\"]TEST(10)

メモ： A) で取得したジャーナルファイル名を B)のクエリ実行時の引数として指定します。

 

B)のクエリで取得できる内容については、クラスリファレンスをご参照ください。
%SYS.Journal:Listクエリの取得内容について
 

A）と B）を組み合わせたメソッド例は以下の通りです。

/// ?1??? ??????????^?????????????
/// ?2??? ????????????????????
ClassMethod SearchJournal(name As %String, filename As %String = "")
{
    if $get(name)="" {
        write "???????????????",!
        quit
    }
    //???????????????????????????
    set stmt=##class(%SQL.Statement).%New()
    set status=stmt.%PrepareClassQuery("%SYS.Journal.File","ByTimeReverseOrder")
    set rs1=stmt.%Execute()
    while rs1.%Next() {
        set jrnfile=rs1.%Get("Name")
        set size=rs1.%Get("Size")
        write "??????????",jrnfile,!," ????????",size/1024/1024," MB",!
        do ..SearchGlo(name,jrnfile)
    }

}

ClassMethod SearchGlo(name As %String, jrnfile As %String)
{
    set stmt=##class(%SQL.Statement).%New()
    set status=stmt.%PrepareClassQuery("%SYS.Journal.Record","List")
    write status,!

    //???????????????????????????????
    set rs1=stmt.%Execute(jrnfile,
"GlobalReference,NewValue",,,$lb("GlobalReference","[",name))
    
    while rs1.%Next() {
        set glo=rs1.%Get("GlobalReference")
        write " ",glo," = ",rs1.%Get("NewValue"),!
    }
}

以下、%ResultSetクラスを利用した実行例です。%SQL.Statementで実行できないバージョンで実行する
場合にご利用ください。

/// ?1??? ??????????^?????????????
/// ?2??? ????????????????????
ClassMethod SearchJournal(name As %String, filename As %String = "")
{
 if $get(name)="" {
  write "???????????????",!
  quit
 }
 //???????????????????????????
 set rs1=##class(%ResultSet).%New()
 set rs1.ClassName="%SYS.Journal.File"
 set rs1.QueryName="ByTimeReverseOrder"
 do rs1.%Execute()
 while rs1.%Next() {
  set jrnfile=rs1.%Get("Name")
  set size=rs1.%Get("Size")
  write "??????????",jrnfile,!," ????????",size/1024/1024," MB",!
  do ..SearchGlo(name,jrnfile)
 }
 do rs1.Close()
}

ClassMethod SearchGlo(name As %String, jrnfile As %String)
{
 set rs1=##class(%ResultSet).%New()
 set rs1.ClassName="%SYS.Journal.Record"
 set rs1.QueryName="List"
 //???????????????????????????????
 do rs1.%Execute(jrnfile,"GlobalReference,NewValue",,,$lb(
"GlobalReference","[",name))
 
 while rs1.%Next() {
  set glo=rs1.%Get("GlobalReference")
  write " ",glo," = ",rs1.%Get("NewValue"),!

 }
 do rs1.Close()
}

 

#グローバル #システム管理 #ヒントとコツ #Caché #Ensemble #InterSystems IRIS #InterSystems IRIS for Health 

 

    ソースURL:
https://jp.community.intersystems.com/post/%E3%82%B3%E3%83%9E%E3%83%B3%E3%83%89%E3%81%A7
%E3%82%B8%E3%83%A3%E3%83%BC%E3%83%8A%E3%83%AB%E3%83%95%E3%82%A1%E3%82%A4%
E3%83%AB%E3%81%AB%E3%81%82%E3%82%8B%E7%89%B9%E5%AE%9A%E3%81%AE%E3%82%B0%E
3%83%AD%E3%83%BC%E3%83%90%E3%83%AB%E5%A4%89%E6%95%B0%E3%82%92%E6%A4%9C%E7
%B4%A2%E3%81%99%E3%82%8B%E6%96%B9%E6%B3%95 

